BIOMETRICS INFORMATION

(You're 95% likely to need this information)

PAMPHLET NO. # 35 DATE: January 8, 1992

SUBJECT: The Computation of Tree Shadow Lengths

The objective of the tree shadow project was to calculate the shadow length of a tree situated on any surface, levelled or sloped, at any location, for any date and at any time. This requires the complex computation of solar altitude and azimuth. The purpose of this pamphlet is to document these computation procedures for use with other studies in forestry.

As a simple example, consider a tree on a south facing slope of M degrees as shown in figure 1 below[1]. If the sun is due south at an altitude of A then by trigonometry, angle

$H = A + M$ and

$L = 90 - A$. If h is the tree height and ℓ is the tree shadow length along the ground, then by the sine law

$$\frac{\ell}{\sin L} = \frac{h}{\sin H}$$

$$\ell = \frac{h \sin(90 - A)}{\sin(A + M)}$$

$$\ell = \frac{h \cos A}{\sin(A + M)} \qquad (1)$$



FIGURE 1. South facing slope of M degrees

If the slope is not directly facing the sun, say with an aspect θ and the solar azimuth is A_z, both measuring clockwise from the north, then the tree shadow will run along a different slope as indicated in figure 2. The new slope B is related to M by the equation:

$$\tan B = \tan M \cos \gamma \qquad (2)$$

where $\gamma = A_z - \theta$, (3)

is the aspect relative to the sun. Therefore the shadow length ℓ is

$$\ell = \frac{h \cos A}{\sin(A + B)} \qquad (4)$$

where B defined as in equation (2) above.



FIGURE 2. Aspect is γ relative to the sun

[1] Special thanks to the Communication & Extension Service Section for producing the two figures on this page.

British Columbia
Ministry of Forests
Research Program

Solar altitude and solar azimuth change with location, date and time. Procedures for determining their values are given by Lee (1978). The solar altitude is calculated by

$$\sin A = \sin \phi \sin \delta + \cos \phi \cos \delta \cos(\omega t) \tag{5}$$

where A = solar altitude in degrees;

δ = solar declination in degrees; $\delta = 23.45 \sin \left[\frac{2\pi \ (284+n)}{365} \right]$

ϕ = terresterial latitude in degrees;

n = Julian date;

A_z = solar azimuth in degrees measured clockwise from north;

t = time in hours from solar noon; eg. t = -2 for 10:00 am, t = 5 for 5:00 pm;

ω = angular velocity of the earth's rotation; $\omega = \frac{\pi}{12}$ rad/hr.

Negative values of A indicate that the sun is below the observer's true horizon. Solar azimuth is given by the equation

$$\sin A_z = \frac{\cos \delta \ \sin(\omega t)}{\cos A} . \tag{6}$$

Negative values of A_z have no physical meaning but occur due to the inverse of the sine function. Since sine is positive in quadrants 1 and 2, negative in quadrants 3 and 4, it is necessary to establish the correct quadrant in which the sun is situated. Kaufmann and Weatherred (1982) suggest comparing ω t with the hour angle, h_w :

If $|\omega t| > h_w$ then if time is at solar noon, that is t = 0, then $A_z = A_z + 180$;

otherwise $A_z = \text{sign}(\omega t) * 180 - A_z$;

where

$$\cos h_w = \frac{\tan \delta}{\tan \phi} . \tag{7}$$

With these values of A and A_z, equation (4) can be used to compute the tree shadow length. As an example, suppose the relationship between tree shadow length (ℓ), slope (M), and aspect (θ) is of interest. If the location is at 50° N latitude (near Quesnel B.C.), and the shadow lengths for August 1st at 12:00 noon are to be computed, then the SAS program shown on the next page calculates the shadow lengths for slopes varying from 0% to 150%, and aspects from 0 to 360 degrees. Figure 3 is a 3-D plot for ℓ, M, and θ. It is truncated at ℓ = 200% in order to reveal the interesting shape near the bottom. Notice that the 3-D graph is symmetric about 0°(N) and 180°(S) aspect. For small slopes, the shadow length versus aspect plots have an approximate "U" shape. However, for large slopes, the plots have a "W" shape. This suggests that slope and aspect are not independent factors. Their effect on shadow lengths cannot be separated.

References:

Merrill R. Kaufmann and James D. Weatherred, 1982, *Determination of Potential Direct Beam Solar Irradiance*, RM-242, U.S. Department of Agriculture, Washington, D.C.

Richard Lee, 1978. *Forest Microclimatology*. Columbia University Press, New York, N.Y.

CONTACT: Vera Sit
356-0435

```
data work;
    t = 0;                                                    /* time set at solar noon */
    pi=4* atan(1);
    rad=pi/180;                                               /* radian to degree */
    deg=1/rad;                                                /* degree to radian */
    phi=50*rad;                                               /* terrestrial latitude = 50 N */
    omega=pi/12;                                              /* angular vel. of the earth */
    nday=juldate(mdy(08,01,00));                              /* Julian date for August 1st */
    delta=23.45*sin(2*pi*(284+nday)/365)*rad;                 /* solar declination*/
    hw = arcos(tan(delta)/tan(phi));                          /* hour angle */
    wt = omega*t;
    a=arsin(sin(phi)*sin(delta) + cos(phi)*cos(delta)*cos(wt)); /*altitude */
    az = arsin(cos(delta)*sin(wt)/cos(a));                    /* azimuth - clockwise from N */
    if (abs(wt)>hw) then do;
         if (t=0) then az=pi-az; else az=sign(wt)*pi-az; end;
    az=az+pi;
    do m = 0 to 150 by 10;                                    /* slope : percent  */
         do ap = 0 to 360 by 10;                              /* aspect : N=0 */
              gamma = az-ap*rad;                              /* corrected aspect */
              beta = atan(m*cos(gamma)/100);                  /* effective slope */
              l = 100 * cos(a) / sin(a+beta);                 /* shadow length : % */
              output;
         end;
    end;
    keep m ap l;
goptions reset=all noprompt gsfmode=replace device=ps chartype=5 border hsize=7 vsize=10;
proc g3d data=work;
    title1 h=2.5 'Shadow Length Plot (percent of tree height)';
    title2 h=1.5 '50 N Latitude on August 1st, 12:00 Noon';
    title3 h=2 angle=90 ' ';
    label ap='Aspect (degrees)' m='Slope (percent)' l='Shadow Length';
    plot ap*m=l/ side rotate=70 tilt=65 zmin=50 zmax=200 zticknum = 6;
run;
/* Note this program was run on the SUN Work station in Research Branch. To run this program
on a PC the GOPTIONS statement must be changed for your printer. */
```

SAS Program for Tree Shadow Lengths Computation



Figure 3: Shadow Length Plot